

2.5利用文本和数字特征对数据分列

路径：D:\Users\35473\Desktop\数据分析学习成果项目纪律记录\从原始数据到分析报告中介绍案例项目\2.5利用文本和数字特征对数据分列

图 1-36 就是一个例子，要从批次中把产品的数量和单位规格分开，但是单位规格不仅仅是汉字，还有数字、括号和字母。

	A	B	C	D	E
1	日期	品名	批次	数量	单位规格
2	2014-2-28	aaaa	120瓶(100ml)		
3	2014-3-1	bbbb	200支(5ml)		
4	2014-3-2	cccc	2200箱		
5	2014-3-3	eeee	20000瓶(50ml)		
6	2014-3-4	dddd	1006箱(AT-250)		

图 1-36 批次是由数量和单位组成，无法进行统计分析

	A	B	C	D	E	F	G
1	日期	品名	批次	数量	单位规格		
2	2014-2-28	aaaa	120瓶(100ml)				
3	2014-3-1	bbbb	200支(5ml)				
4	2014-3-2	cccc	2200箱				
5	2014-3-3	eeee	20000瓶(50ml)				
6	2014-3-4	dddd	1006箱(AT-250)				
7							
8							
9							
10							



知识点

当字符与数字混合时！？怎么办？

思路

先用MID函数把批次的每个字符提取出来！生成一个数组！将这个数组的每个元素都乘1，结果如果是错误值，表明是文字！结果是数字，表明是数字！

之后就可以利用MATCH函数判断出现错误的位置了！

Warning

这里是一个数组公式！，按 **CTRL+shift+enter**

介绍一下 `match`(要查找的值, 查找区域, 匹配方式 (默认1)) 函数
查找某个值在指定区域, 并返回相对位置, 从1开始计数

记住一个固定公式:

从单元格 C2 的文本开头提取出连续的数字部分, 并将其转换为真正的数值

`indirect("1:" & len())`: 将文本转换为excel可以理解的单元格区域引用: 1:6 第一行到第六行

它返回的是一个引用, 必须被ROW进一步处理才可以产生实际值

`ROW(indirect("1:" & len()))` 返回引用区域中每一行的行号!

当参数为一个多行的区域引用时, ROW会返回一个垂直数组

`MID(C2, ROW(indirect("1:" & len())), 1)`: 提取字符

`ISERROR(数组)` 判断数组中的每个元素是否为任何错误

返回逻辑值数组

match会查找第一个出现的值!

1 * 数字字符串 : 可以将字符转换为数字



数量

```
=1*left(C2,match(true,iserror(1*mid(C2,row(indirect("1:" & len(C2))),1)),0)-1)
```

	A	B	C	D	E	F
1	日期	品名	批次	数量	单位规格	
2	2014-2-28	aaaa	120瓶(100ml)	120		
3	2014-3-1	bbbb	200支(5ml)	200		
4	2014-3-2	cccc	2200箱	2200		
5	2014-3-3	eeee	20000瓶(50ml)	20000		
6	2014-3-4	dddd	1006箱(AT-250)	1006		
7						
8						

单位规格

```
=right(C2,len(C2)-len(D2))
```

	A	B	C	D	E	F
1	日期	品名	批次	数量	单位规格	
2	2014-2-28	aaaa	120瓶(100ml)	120	瓶(100ml)	
3	2014-3-1	bbbb	200支(5ml)	200	支(5ml)	
4	2014-3-2	cccc	2200箱	2200	箱	
5	2014-3-3	eeee	20000瓶(50ml)	20000	瓶(50ml)	
6	2014-3-4	dddd	1006箱(AT-250)	1006	箱(AT-250)	
7						
8						
9						